

# NAG Fortran Library Routine Document

## G05EHF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

G05EHF performs a pseudo-random permutation of a vector of integers.

### 2 Specification

```
SUBROUTINE G05EHF (INDEX, N, IFAIL)
  INTEGER          INDEX(N), N, IFAIL
```

### 3 Description

The routine permutes the elements of INDEX without inspecting their values. Each of the  $n!$  possible permutations of the  $n$  values may be regarded as being equally probable.

Even for modest values of  $n$  (greater than 25 say), it is theoretically impossible that all  $n!$  permutations may occur, as  $n!$  exceeds the cycle length of G05CAF. For practical purposes this is irrelevant, as the time necessary to generate all possible permutations is many millenia.

### 4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison-Wesley

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin

### 5 Parameters

- 1: INDEX(N) – INTEGER array *Input/Output*  
*On entry:* the  $n$  integer values to be permuted.  
*On exit:* the  $n$  permuted integer values.
- 2: N – INTEGER *Input*  
*On entry:* the number of values to be permuted.  
*Constraint:*  $N \geq 1$ .
- 3: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry,  $N < 1$ .

## 7 Accuracy

Not applicable.

## 8 Further Comments

The time taken by the routine is of order  $n$ .

In order to permute other kinds of vectors, or matrices of higher dimension, the following technique may be used:

- (a) Set  $INDEX(i) = i$ , for  $i = 1, 2, \dots, n$
- (b) Use G05EHF to permute INDEX
- (c) Use the contents of INDEX as a set of indices to access the relevant vector or matrix.

In order to divide pseudo-randomly a vector or matrix into subgroups of chosen sizes, a similar procedure may be used. INDEX is first set to the number of 1s, 2s, etc., corresponding to the size of each group, then permuted, and used to index the groups.

## 9 Example

A vector containing the first eight positive integers in ascending order is permuted and the permutation is printed. This is repeated a total of ten times.

The generator mechanism used is selected by an initial call to G05ZAF.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G05EHF Example Program Text
*      Mark 20 Revised. NAG Copyright 2001.
*      .. Parameters ..
      INTEGER                N
      PARAMETER              (N=8)
      INTEGER                NOUT
      PARAMETER              (NOUT=6)
*      .. Local Scalars ..
      INTEGER                I, IFAIL, J, K, M
*      .. Local Arrays ..
      INTEGER                INDEX(N)
*      .. External Subroutines ..
      EXTERNAL               G05CBF, G05EHF, G05ZAF
*      .. Executable Statements ..
      CALL G05ZAF('O')
      WRITE (NOUT,*) 'G05EHF Example Program Results'
      WRITE (NOUT,*)
      M = 10
      CALL G05CBF(0)
      WRITE (NOUT,99998) M, ' Permutations of first ', N, ' integers'
      WRITE (NOUT,*)
      DO 40 J = 1, M
```

```
      DO 20 I = 1, N
        INDEX(I) = I
20    CONTINUE
      IFAIL = 0
*
      CALL G05EHF(INDEX,N,IFAIL)
*
      WRITE (NOUT,99999) (INDEX(K),K=1,N)
40  CONTINUE
      STOP
*
99999 FORMAT (1X,8I3)
99998 FORMAT (1X,I2,A,I1,A)
      END
```

## 9.2 Program Data

None.

## 9.3 Program Results

G05EHF Example Program Results

10 Permutations of first 8 integers

7	8	1	2	4	6	3	5
3	1	4	6	7	8	5	2
7	6	5	1	3	4	8	2
6	2	7	3	8	5	1	4
1	6	2	4	7	8	5	3
4	1	5	8	7	6	3	2
8	3	1	6	4	2	5	7
1	2	6	7	8	4	3	5
2	5	7	6	3	1	4	8
2	8	6	7	3	5	1	4

---